

1 SUMMARY

To calculate the **best weighted least squares fit by a sum of n decaying exponentials**. It finds parameters a_j and $b_j, j=1,2,..,n$ which minimize

$$S = \sum_{i=1}^m w_i^2 \left(y_i - \sum_{j=1}^n a_j e^{b_j x_i} \right)^2, \quad m \geq 2n, b_j \leq 0, j=1,2,..,n.$$

The subroutine utilizes both first and second derivatives and calculates these itself.

The user must specify the accuracy required for the parameters b_j . The subroutine returns the parameter variance-covariance matrix, the sum of squares and the residuals.

ATTRIBUTES — **Version:** 1.0.0. **Types:** VC15A; VC15AD. **Calls:** MA44 and VA21. **Original date:** March 1985. **Origin:** M.D.Hebden/J.K.Reid, Harwell. **Remark:** This is a slightly rewritten version of VC05 and supersedes it.

2 HOW TO USE THE PACKAGE

2.1 Argument lists

The single precision version

```
CALL VC15A(X, Y, W, M, B, N, EPS, IPRINT, MAXFN, LP, WORK, IWORK)
```

The double precision version

```
CALL VC15AD(X, Y, W, M, B, N, EPS, IPRINT, MAXFN, LP, WORK, IWORK)
```

- X is a REAL (DOUBLE PRECISION in the D version) array of length m which the user must set to hold the positions of the data points $x_i, i=1,2,..,m$. It is not altered.
- Y is a REAL (DOUBLE PRECISION in the D version) array of length m which the user must set to hold the data values $y_i, i=1,2,..,m$. It is not altered.
- W is a REAL (DOUBLE PRECISION in the D version) array of length m which the user must set to hold the weights $w_i, i=1,2,..,m$. Note that the problem in which all the data is known to the same relative accuracy can be handled by setting $w_i=1/y_i, i=1,2,..,m$ and the problem in which all the data is known to the same absolute accuracy can be handled by setting $w_i=1, i=1,2,..,m$. W is not altered.
- M is an INTEGER variable which must be set by the user to m , the number of data points. It is not altered. **Restriction:** $m \geq 2n$.
- B is a REAL (DOUBLE PRECISION in the D version) array of length $2n$ which is used to hold the parameters. Before entry the user must set $B(j), j=1,2,..,n$, to initial estimates of the parameters b_j . On return these array elements will hold the final estimates of these parameters, in ascending order. $B(n+j), j=1,2,..,n$ need not be set on entry and on return contain the final values of the parameters a_j . **Restriction:** $b_j \leq 0, j=1,2,..,n$.
- N is an INTEGER variable which must be set by the user to n , the number of exponential terms to be fitted. The problem is likely to be very ill-conditioned if n is greater than about 3 or 4, which means that it may not be possible to find accurate parameter values. **Recommendation:** $n \leq 4$. **Restriction:** $n \geq 1$.
- EPS is a REAL (DOUBLE PRECISION in the D version) array of length n which the user must set to the required absolute accuracy in the exponential parameters $b_j, j=1,2,..,n$. It is not altered.
- IPRINT is an INTEGER variable which must be set by the user to control the printing produced by VA21, which is called by VC15. If IPRINT=0 no printing is done; otherwise printing takes place every |IPRINT| iterations and

before return. The iteration number, number of function evaluations, $f(\mathbf{b}) = \min S(\mathbf{a}, \mathbf{b})$ and the current value of \mathbf{b} (called \mathbf{x} by VA21) are printed, and if IPRINT<0 the first and second derivatives of f are printed too. IPRINT is not altered.

MAXFN is an INTEGER variable which must be set by the user to a limit on the number of times the sum of squares is to be calculated. It is not altered.

LP is an INTEGER variable which must be set by the user to a unit number for printing or to a non-positive number if printing is to be suppressed.

WORK is a REAL (DOUBLE PRECISION in the D version) array of length

$$\max(2+8n+4n^2+2m+2mn, 2+12n+3n^2+4m+mn),$$

which is used as workspace. On return

- (1) WORK(1) contains the final sum of squares S .
- (2) WORK(2) to WORK($4n^2+1$) contains the variance-covariance matrix corresponding to the vector (\mathbf{b}, \mathbf{a}) .
- (3) WORK($4n^2+2$) to WORK($4n^2+m+1$) contains the residuals.
- (4) WORK($4n^2+m+2$) contains an error flag. It is set positive on a normal return and to a negative value if the results are unreliable.

IWORK is an INTEGER workarray of length $2n+1$.

3 GENERAL INFORMATION

Workspace: Provided in the argument WORK and IWORK.

Use of common: None.

Other routines called directly: MA44, VA21 and the private subroutines VC15Y/YD and VC15Z/ZD are called.

Input/output: Output is under the control of arguments IPRINT and LP.

Restrictions: $m \geq 2n$, $n \geq 1$, $b_j \leq 0$ ($j=1,2,\dots,n$).

4 METHOD

Given an estimate of the non-linear parameters b_i , $i=1,2,\dots,n$, a best set of linear parameters a_i $i=1,2,\dots,n$ is obtained using MA44A/AD. In this way the sum of squares can be regarded as a function only of the non-linear parameters b_i , $i=1,2,\dots,n$. VA21A/AD, which requires both first and second derivatives, is used to minimize this sum of squares.

The code is a minor modification of the code VC05 of Hebden (1973). The modifications bring the code into line with Fortran 77, give more control over the printing, and make the printing clearer. Also MA44 is called for calculating the estimated variance-covariance matrix; this is more reliable than VC05's explicit formation and inversion of the normal matrix.

5 EXAMPLE OF USE

As a very simple example, the following code fits a sum of two exponentials to the data 2.0000, 1.3679, 1.1353, 1.0498, 1.0183, 1.0067, 1.0025, 1.0009, 1.0003, 1.0001, 1.0000 at the points 0,1,2,...,10 with weights unity.

```

DOUBLE PRECISION X(11),Y(11),W(11),B(4),EPS(2),WORK(300)
INTEGER IWORK(5)
DATA X/0.0000D0,1.0000D0,2.0000D0,3.0000D0,4.0000D0,5.0000D0,
* 6.0000D0,7.0000D0,8.0000D0,9.0000D0,10.0000D0/
DATA Y/2.0000D0,1.3679D0,1.1353D0,1.0498D0,1.0183D0,1.0067D0,
* 1.0025D0,1.0009D0,1.0003D0,1.0001D0,1.0000D0/
DATA W/11*1.0000D0/
DATA EPS/2*1.0D-4/
M=11
B(1)=-0.1D0
B(2)=-0.9D0
N=2
IPRINT=4
MAXFN=20
LP=6
CALL VC15AD(X,Y,W,M,B,N,EPS,IPRINT,MAXFN,LP,WORK,IWORK)
WRITE (LP,160) WORK(1)
160 FORMAT (' OPTIMAL WEIGHTED SUM OF SQUARES IS', D15.7)
WRITE (LP,170)
170 FORMAT (' OPTIMAL PARAMETERS ARE')
WRITE (LP,180)
180 FORMAT (' I      A(I)          ST. DEV.          B(I)          ',
* 'ST. DEV.')
WRITE(LP,190) (I,B(N+I),DSQRT(WORK(2+(I+N-1)*(2*N+1))),
* B(I),DSQRT(WORK(2+(I-1)*(2*N+1))),I=1,N)
190 FORMAT (I4, 1P, 4D13.5)
STOP
END

```

This produces the following output

```

ENTRY TO VA21A

ITERATION   1,      1 FUNCTION EVALUATIONS, F =  5.3940033D-01
X = -9.0000000D-01 -1.0000000D-01

ITERATION   5,      5 FUNCTION EVALUATIONS, F =  6.6809495D-07
X = -9.9815643D-01 -1.7751159D-05

ITERATION   6,      6 FUNCTION EVALUATIONS, F =  4.3568702D-09
X = -9.9998831D-01 -3.5994569D-06
OPTIMAL WEIGHTED SUM OF SQUARES IS  0.4356870D-08
OPTIMAL PARAMETERS ARE
 I      A(I)          ST. DEV.          B(I)          ST. DEV.
 1  1.00000D 00  4.23992D-05 -9.99974D-01  9.40255D-05
 2  9.99999D-01  3.66421D-05 -3.07758D-06  4.93451D-06

```